

Rate-equation models for a nitrogen molecule (N₂) in LCLS radiation

Christian Buth

Argonne National Laboratory, Argonne, Illinois 60439, USA

Max-Planck-Institut für Kernphysik, Saupfercheckweg 1, 69117 Heidelberg, Germany

The PULSE Institute for Ultrafast Energy Science, SLAC National Accelerator Laboratory, Menlo Park, California 94025, USA

Ji-Cai Liu

Max-Planck-Institut für Kernphysik, Saupfercheckweg 1, 69117 Heidelberg, Germany

19 April 2012

Initialize package

```
BeginPackage["NitrogenMoleculeRateEqs`"]
```

```
Needs["NitrogenRateEquations`"]
```

Public section

```
NitrogenMoleculeRateEqs::usage="NitrogenMoleculeRateEqs implements a rate-equation " <>  
"model for the interaction of x rays with a single nitrogen molecule (N2)."
```

Crater profile

FWHM of the crater profile:

```
rhoXx:=2.2*MicrometerToBohr
rhoYy:=1.2*MicrometerToBohr
```

The crater is normalized to unity with respect to an integration over the whole xy plane:

```
craterX[rx_] := 1/rhoXx/Sqrt[Pi/Log[16]] * Exp[-4*Log[2] * (rx/rhoXx)^2]
craterY[ry_] := 1/rhoYy/Sqrt[Pi/Log[16]] * Exp[-4*Log[2] * (ry/rhoYy)^2]
crater[rx_,ry_] := craterX[rx] * craterY[ry]
```

Number of radial integration points for the crater integration:

```
nintdata:=100
```

Fluence profile

Number of x-ray photons for "monochromatic" light pulses:

```
numXRays[pulseenergy_,photonenergy_] := pulseenergy/photonenergy
```

X-ray fluence, i.e., the number of photos per squaremeter:

```
fluence[pulseenergy_,photonenergy_,rx_,ry_] := numXRays[pulseenergy,photonenergy] * crater[rx,ry]
```

```
peakfluence[pulseenergy_,photonenergy_] := fluence[pulseenergy,photonenergy,0,0]
```

The peak x-ray flux at a point in the beam area:

```
J0Xfun[pulseenergy_,photonenergy_,tauX_,rx_,ry_] := 2*Sqrt[Log[2]/Pi] / tauX * fluence[pulseenergy,photonenergy,rx,ry]
```

Gaussian x-ray flux pluse shape:

```
jxgaussian[J0Xj_,tX_,t_] := J0Xj * Exp[-4*Log[2] * (t/tX)^2] * UnitStep[t+2*tX] * UnitStep[2*tX-t]
```

Crater integration of ion yields from a single nitrogen atom

All charge states: N, N⁺, N⁽²⁺⁾, N⁽³⁺⁾, N⁽⁴⁺⁾, N⁽⁵⁺⁾, N⁽⁶⁺⁾, N⁽⁷⁺⁾:

```
ionyields[sol_] := {solGround[sol, posinfinity], solSingly[sol, posinfinity], solDoubly[sol, posinfinity],
  solTriply[sol, posinfinity], solQuadruply[sol, posinfinity], solQuintuply[sol, posinfinity],
  solSextuply[sol, posinfinity], solSeptuply[sol, posinfinity]}
```

Compute ion yields for a fraction of the peak x-ray fluences on the beam axis:

```
ionprofile[pulseenergy_, photonenergy_, tauX_, nintsteps_, nint_, truth_] := Module[{J0Xpeak, J0Xcur, sol, jxcrater},
  J0Xpeak = 2*Sqrt[Log[2]/Pi]/tauX*peakfluence[pulseenergy, photonenergy];
  J0Xcur = J0Xpeak/nintsteps*nint;
  jxcrater[t_] := jxgaussian[J0Xcur, tauX, t];
  sol = solveNitrogenRateEquations[photonenergy, jxcrater, neginfinity, posinfinity, 1/1000, truth];
  Join[{J0Xcur}, ionyields[sol]]
]
```

```
tabionprof[pulseenergy_, photonenergy_, tauX_, nintsteps_, truth_] := Module[{nint},
  Table[ionprofile[pulseenergy, photonenergy, tauX, nintsteps, nint, truth],
  {nint, nintsteps, 0, -1}]
]
```

Create interpolation functions for all charge states: N, N⁺, N⁽²⁺⁾, N⁽³⁺⁾, N⁽⁴⁺⁾, N⁽⁵⁺⁾, N⁽⁶⁺⁾, N⁽⁷⁺⁾:

```
interpolYields[tabdata_] := Module[{nint},
  Table[
  ListInterpolation[tabdata[[All, nint]], {tabdata[[All, 1]]}],
  {nint, 2, Length[tabdata[[1]]}]
]
```

Evaluate all ion yields at a specific spatial point:

```
ionyieldint[funyields_, pulseenergy_, photonenergy_, tauX_, rx_, ry_] := Module[{jj0x, nint},
  jj0x = J0Xfun[pulseenergy, photonenergy, tauX, rx, ry];
  Table[funyields[[nint]][jj0x], {nint, 1, Length[funyields]}]
]
```

Integration of the ion yields of the beam area. The integral needs to be normalized to the integration area:

```

ionyieldbeam[funyields_,pulseenergy_,photonenergy_,tauX_]:=Module[{iytot},
Off[Solve::ifun];
Off[NIntegrate::slwcon];
iytot=1/((2*5*MicrometerToBohr)^2)*NIntegrate[ionyieldint[funyields,
pulseenergy,photonenergy,tauX,rx,ry],{rx,-5*MicrometerToBohr,
5*MicrometerToBohr},{ry,-5*MicrometerToBohr,5*MicrometerToBohr},MaxRecursion->10000,
Method->"LocalAdaptive"];
On[Solve::ifun];
On[NIntegrate::slwcon];
iytot
]

```

```

ionyieldcrateranalysis[pulseenergy_,photoenergy_,pulseduration_,nint_,
inclDCH_]:=Module[{tabionyi,funionyi,crationyi},
tabionyi=tabionprof[pulseenergy,photoenergy,pulseduration,nint,inclDCH];
funionyi=interpolYields[tabionyi];
crationyi=ionyieldbeam[funionyi,pulseenergy,photoenergy,pulseduration]
]

```

Independent-atom nitrogen molecule

Probabilities to find N₂ in the possible molecular charge states:

```
imolGround[sol_,t1_]:=solGround[sol,t1]^2
```

```
imolOne[sol_,t1_]:=2*solGround[sol,t1]*solSingly[sol,t1]
```

```
imolTwo[sol_,t1_]:=2*solGround[sol,t1]*solDoubly[sol,t1]+
solSingly[sol,t1]^2
```

```
imolThree[sol_,t1_]:=2*solGround[sol,t1]*solTriply[sol,t1]+
2*solSingly[sol,t1]*solDoubly[sol,t1]
```

```
imolFour[sol_,t1_]:=2*solGround[sol,t1]*solQuadruply[sol,t1]+
2*solSingly[sol,t1]*solTriply[sol,t1]+
solDoubly[sol,t1]^2
```

```
imolFive[sol_,t1_]:=2*solGround[sol,t1]*solQuintuply[sol,t1]+
2*solSingly[sol,t1]*solQuadruply[sol,t1]+
2*solDoubly[sol,t1]*solTriply[sol,t1]
```

```
imolSix[sol_,t1_]:=2*solGround[sol,t1]*solSextuply[sol,t1]+
  2*solSingly[sol,t1]*solQuintuply[sol,t1]+
  2*solDoubly[sol,t1]*solQuadruply[sol,t1]+
  solTriply[sol,t1]^2
```

```
imolSeven[sol_,t1_]:=2*solGround[sol,t1]*solSeptuply[sol,t1]+
  2*solSingly[sol,t1]*solSextuply[sol,t1]+
  2*solDoubly[sol,t1]*solQuintuply[sol,t1]+
  2*solTriply[sol,t1]*solQuadruply[sol,t1]
```

```
imolEight[sol_,t1_]:=2*solSingly[sol,t1]*solSeptuply[sol,t1]+
  2*solDoubly[sol,t1]*solSextuply[sol,t1] +
  2*solTriply[sol,t1]*solQuintuply[sol,t1]+
  solQuadruply[sol,t1]^2
```

```
imolNine[sol_,t1_]:=2*solDoubly[sol,t1]*solSeptuply[sol,t1]+
  2*solTriply[sol,t1]*solSextuply[sol,t1]+
  2*solQuadruply[sol,t1]*solQuintuply[sol,t1]
```

```
imolTen[sol_,t1_]:=2*solTriply[sol,t1]*solSeptuply[sol,t1]+
  2*solQuadruply[sol,t1]*solSextuply[sol,t1]+
  solQuintuply[sol,t1]^2
```

```
imolEleven[sol_,t1_]:=2*solQuadruply[sol,t1]*solSeptuply[sol,t1]+
  2*solQuintuply[sol,t1]*solSextuply[sol,t1]
```

```
imolTwelve[sol_,t1_]:=2*solQuintuply[sol,t1]*solSeptuply[sol,t1]+
  solSextuply[sol,t1]^2
```

```
imolThirteen[sol_,t1_]:=2*solSextuply[sol,t1]*solSeptuply[sol,t1]
```

```
imolFourteen[sol_,t1_]:=solSeptuply[sol,t1]^2
```

All charge states for each atom N, N⁺, N⁽²⁺⁾, N⁽³⁺⁾, N⁽⁴⁺⁾, N⁽⁵⁺⁾, N⁽⁶⁺⁾, N⁽⁷⁺⁾:

```
imolIonyields[sol_]:={imolGround[sol, posinfinity], imolOne[sol, posinfinity], imolTwo[sol, posinfinity],
  imolThree[sol, posinfinity], imolFour[sol, posinfinity], imolFive[sol, posinfinity],
  imolSix[sol, posinfinity], imolSeven[sol, posinfinity], imolEight[sol, posinfinity],
  imolNine[sol, posinfinity], imolTen[sol, posinfinity], imolEleven[sol, posinfinity],
  imolTwelve[sol, posinfinity], imolThirteen[sol, posinfinity], imolFourteen[sol, posinfinity]}
```

Convert molecular ion yields to atomic ion yields:

```

imolatatomicprob[imolprob_] := {
  imolprob[[1]] + 1/2*imolprob[[2]],
  1/2*imolprob[[2]] + imolprob[[3]] + 1/2*imolprob[[4]],
  1/2*imolprob[[4]] + imolprob[[5]] + 1/2*imolprob[[6]],
  1/2*imolprob[[6]] + imolprob[[7]] + 1/2*imolprob[[8]],
  1/2*imolprob[[8]] + imolprob[[9]] + 1/2*imolprob[[10]],
  1/2*imolprob[[10]] + imolprob[[11]] + 1/2*imolprob[[12]],
  1/2*imolprob[[12]] + imolprob[[13]] + 1/2*imolprob[[14]],
  1/2*imolprob[[14]] + imolprob[[15]]
}

```

Compute ion yields for a fraction of the peak x-ray fluences on the beam axis:

```

imolionprofile[pulseenergy_, photonenergy_, tauX_, nintsteps_, nint_, truth_] := Module[
  {J0Xpeak, J0Xcur, sol, jxcrater},
  J0Xpeak = 2*Sqrt[Log[2]/Pi]/tauX*peakfluence[pulseenergy, photonenergy];
  J0Xcur = J0Xpeak/nintsteps*nint;
  jxcrater[t_] := jxgaussian[J0Xcur, tauX, t];
  sol = solveNitrogenRateEquations[photonenergy, jxcrater, neginfinity, posinfinity, 1/1000, truth];
  Join[{J0Xcur}, imolionyields[sol]]
]

```

```

imoltabionprof[pulseenergy_, photonenergy_, tauX_, nintsteps_, truth_] := Table[
  imolionprofile[pulseenergy, photonenergy, tauX, nintsteps, nint, truth], {nint, nintsteps, 0, -1}
]

```

Create interpolation functions for all charge states: N, N⁺, N⁽²⁺⁾, ..., N⁽¹⁴⁺⁾:

```

imolinterpolYields[tabdata_] := Module[{nint},
  Table[
  ListInterpolation[tabdata[[All, nint]], {tabdata[[All, 1]}], Method -> "Spline",
  {nint, 2, 16}]
]

```

Evaluate all ion yields at a specific spatial point:

```

imolionyieldsint[funyields_, pulseenergy_, photonenergy_, tauX_, rx_, ry_] := Module[{jj0x, nint},
  jj0x = J0Xfun[pulseenergy, photonenergy, tauX, rx, ry];
  Table[funyields[[nint]][jj0x], {nint, 1, 15}]
]

```

Integration of the ion yields of the beam area. The integral needs to be normalized to the integration area:

```

imolionyieldbeam[funyields_,pulseenergy_,photonenergy_,tauX_]:=Module[{iytot},
Off[Solve::ifun];
Off[NIntegrate::slwcon];
iytot=1/((2*5*MicrometerToBohr)^2)*NIntegrate[imolionyieldint[funyields,
pulseenergy,photonenergy,tauX,rx,ry],{rx,-5*MicrometerToBohr,
5*MicrometerToBohr},{ry,-5*MicrometerToBohr,5*MicrometerToBohr}];
On[Solve::ifun];
On[NIntegrate::slwcon];
iytot
]

```

```

imolionyieldcrateranalysis[pulseenergy_,photoenergy_,pulseduration_,nint_,inclDCH_]:=Module[
{tabionyi,funionyi,crationyi},
tabionyi=imoltabionprof[pulseenergy,photoenergy,pulseduration,nint,inclDCH];
funionyi=imolinterpolYields[tabionyi];
crationyi=imolionyieldbeam[funionyi,pulseenergy,photoenergy,pulseduration]
]

```

Fragmentation matrix model

Volume integration with a fragmentation matrix

All charge states: N, N⁺, N⁽²⁺⁾, N⁽³⁺⁾, N⁽⁴⁺⁾, N⁽⁵⁺⁾, N⁽⁶⁺⁾, N⁽⁷⁺⁾:

```

atomicprobs[sol_]:= {solGround[sol, posinfinity], solSingly[sol, posinfinity], solDoubly[sol, posinfinity],
solTriply[sol, posinfinity], solQuadruply[sol, posinfinity], solQuintuply[sol, posinfinity],
solSextuply[sol, posinfinity], solSeptuply[sol, posinfinity]}

```

```

ics[i_, j_]:=i+1+8*j

```

```

molprobs[fmat_, sol_]:=Module[{i, j, atmps, dirps, molps},
atmps=atomicprobs[sol];
dirps=Flatten[Table[atmps[[i+1]]*atmps[[j+1]], {j, 0, 7}, {i, 0, 7}]];
molps=fmat.dirps;
(* Second Table is a sanity check; both must yield the same result *)
{Table[Sum[molps[[ics[i, j]]], {i, 0, 7}], {j, 0, 7}], Table[Sum[molps[[ics[j, i]]], {i, 0, 7}], {j, 0, 7}]}
]

```

Compute ion yields for a fraction of the peak x-ray fluences on the beam axis:

```
stationprofile[fmat_,pulseenergy_,photonenergy_,tauX_,nintsteps_,nint_,truth_] := Module[{J0Xpeak,J0Xcur,sol,jxcrater},
J0Xpeak=2*Sqrt[Log[2]/Pi]/tauX*peakfluence[pulseenergy,photonenergy];
J0Xcur=J0Xpeak/nintsteps*nint;
jxcrater[t_] := jxgaussian[J0Xcur,tauX,t];
sol=solveNitrogenRateEquations[photonenergy,jxcrater,neginfinity,posinfinity,1/1000,truth];
Join[{J0Xcur},molprobs[fmat,sol][[1]]]
]
```

```
stattabionprof[fmat_,pulseenergy_,photonenergy_,tauX_,nintsteps_,truth_] := Module[{nint},
Table[stationprofile[fmat,pulseenergy,photonenergy,tauX,nintsteps,nint,truth],
{nint,nintsteps,0,-1}]
]
```

Create interpolation functions for all charge states: N, N⁺, N⁽²⁺⁾, N⁽³⁺⁾, N⁽⁴⁺⁾, N⁽⁵⁺⁾, N⁽⁶⁺⁾, N⁽⁷⁺⁾:

```
statinterpolYields[tabdata_] := Module[{nint},
Table[
ListInterpolation[tabdata[[All,nint]],{tabdata[[All,1]]}(*,Method->"Spline"*),
{nint,2,Length[tabdata[[1]]}]
]
```

Evaluate all ion yields at a specific spatial point:

```
stationyeldint[funyields_,pulseenergy_,photonenergy_,tauX_,rx_,ry_] := Module[{jj0x,nint},
jj0x=J0Xfun[pulseenergy,photonenergy,tauX,rx,ry];
Table[funyields[[nint]][jj0x],{nint,1,Length[funyields]}]
]
```

Integration of the ion yields of the beam area. The integral needs to be normalized to the integration area:

```
stationyeldbeam[funyields_,pulseenergy_,photonenergy_,tauX_] := Module[{iytot},
Off[Solve::ifun];
Off[NIntegrate::slwcon];
iytot=1/(2*5*MicrometerToBohr)^2*NIntegrate[stationyeldint[funyields,
pulseenergy,photonenergy,tauX,rx,ry],{rx,-5*MicrometerToBohr,
5*MicrometerToBohr},{ry,-5*MicrometerToBohr,5*MicrometerToBohr},MaxRecursion->10000,
Method->"LocalAdaptive"];
On[Solve::ifun];
On[NIntegrate::slwcon];
iytot
]
```



```
stationyieldcrateranalysis[fmat_,pulseenergy_,photoenergy_,pulseduration_,nint_,
inclDCH_] := Module[{tabionyi,funionyi,crationyi},
tabionyi=stattabionprof[fmat,pulseenergy,photoenergy,pulseduration,nint,inclDCH];
funionyi=statinterpolYields[tabionyi];
crationyi=stationyieldbeam[funionyi,pulseenergy,photoenergy,pulseduration]
]
```

Transformation matrix

```
mixingmat[id1_,id2_,re1_,re2_,fac_] := Module[{mima},
mima=IdentityMatrix[64];
mima[[id1,id1]]=1-fac;
mima[[id2,id2]]=1-fac;
mima[[re1,id1]]=fac;
mima[[re2,id2]]=fac;
mima
]
```

```
icsp[p_] := ics[p[[1]],p[[2]]]
revp[p_] := {p[[2]],p[[1]]}
```

```
icsf1[p_] := If[EvenQ[p[[1]]+p[[2]]],
icsp[{(p[[1]]+p[[2]])/2,(p[[1]]+p[[2]])/2}],
icsp[{(p[[1]]+p[[2]]-1)/2,(p[[1]]+p[[2]]+1)/2}]
]
```

```
icsfr[p_] := If[EvenQ[p[[1]]+p[[2]]],
icsp[{(p[[1]]+p[[2]])/2,(p[[1]]+p[[2]])/2}],
icsp[{(p[[1]]+p[[2]]+1)/2,(p[[1]]+p[[2]]-1)/2}]
]
```

```
genTrafMatrices[] := Module[{redpairs,compairs,i},
redpairs=Subsets[Table[i,{i,0,7}],{2}];
compairs=Select[redpairs,Abs[#[[1]]-#[[2]]]>=2&];
Table[mixingmat[icsp[compairs[[i]]],icsp[revp[compairs[[i]]]],
icsf1[compairs[[i]]],icsfr[compairs[[i]]],ToExpression["f"<>ToString[i]]],
{i,1,Length[compairs]}]
]
```

Compound final fragmentation matrix via

```
trafmats=genTrafMatrices[];
fmat=IdentityMatrix[64];
For[i=1,i<=Length[trafmats],i++,
fmat=fmat.trafmats[[i]];
```

```
]
fmatnom4fs=fmat/.Last[fit4fs];
```

SCHs, DCHs, and PAPAs

Probabilities to find N2 in SCH-, DCH-, and PAPA-channels

```
solPAP[sol_,t1_]:= (Replace[Ppap121v223[t],sol[[1]]]+Replace[Ppap103v223[t],sol[[1]]] +
  Replace[Ppap112v223[t],sol[[1]]]+Replace[Ppap221v123[t],sol[[1]]] +
  Replace[Ppap203v123[t],sol[[1]]]+Replace[Ppap212v123[t],sol[[1]]])/.{t->t1}
```

```
solPAPprob[sol_,t1_]:=Replace[Ppap[t],sol[[1]]]/.{t->t1}
```

```
solPAPA[sol_,t1_]:=Replace[Ppapa[t],sol[[1]]]/.{t->t1}
```

```
solNeutral[sol_,t1_]:= (Replace[P223[t],sol[[1]]]/.{t->t1})^2
```

```
solSCH[sol_,t1_]:= (2*Replace[P223[t],sol[[1]]]*Replace[P123[t],sol[[1]]])/.{t->t1}
```

```
solSCHdecay[sol_,t1_]:=Replace[Pschd[t],sol[[1]]]/.{t->t1}
```

```
solDCHsinglesite[sol_,t1_]:= (2*Replace[P223[t],sol[[1]]] *
  Replace[P023[t],sol[[1]]])/.{t->t1}
```

```
solDCHsinglesiteDecay[sol_,t1_]:=Replace[Pssdchd[t],sol[[1]]]/.{t->t1}
```

```
solDCHtwosites[sol_,t1_]:= (Replace[P123[t],sol[[1]]]/.{t->t1})^2
```

```
solDCHtwositesDecay[sol_,t1_]:=Replace[Ptsdchd[t],sol[[1]]]/.{t->t1}
```

Molecular rate equations

A PA cycle:

```

reqPA221v223[en_, jxpulse_] := Ppa221v223'[t] == (-fsigTot221[en] - fsigTot223[en]) *
    Ppa221v223[t] * jxpulse[t] + GamA221v123*2*P123[t] * P223[t]
ieqPA221v223[neginf_, inival_] := Ppa221v223[neginf] == inival

reqPA203v223[en_, jxpulse_] := Ppa203v223'[t] == (-fsigTot203[en] - fsigTot223[en]) *
    Ppa203v223[t] * jxpulse[t] + GamA203v123*2*P123[t] * P223[t]
ieqPA203v223[neginf_, inival_] := Ppa203v223[neginf] == inival

reqPA212v223[en_, jxpulse_] := Ppa212v223'[t] == (-fsigTot212[en] - fsigTot223[en]) *
    Ppa212v223[t] * jxpulse[t] + GamA221v123*2*P123[t] * P223[t]
ieqPA212v223[neginf_, inival_] := Ppa212v223[neginf] == inival

```

One-site PAPAs:

```

reqPAP121v223[en_, jxpulse_, truth_] := Ppap121v223'[t] == (fsig121v221[en] * Ppa221v223[t] -
    (fsigTot121[en, truth] + fsigTot223[en]) * Ppap121v223[t]) * jxpulse[t] -
    GamATot121 * Ppap121v223[t]
ieqPAP121v223[neginf_, inival_] := Ppap121v223[neginf] == inival

reqPAP103v223[en_, jxpulse_, truth_] := Ppap103v223'[t] == (fsig103v203[en] * Ppa203v223[t] -
    (fsigTot103[en, truth] + fsigTot223[en]) * Ppap103v223[t]) * jxpulse[t] -
    GamATot103 * Ppap103v223[t]
ieqPAP103v223[neginf_, inival_] := Ppap103v223[neginf] == inival

reqPAP112v223[en_, jxpulse_, truth_] := Ppap112v223'[t] == (fsig112v212[en] * Ppa212v223[t] -
    (fsigTot112[en, truth] + fsigTot223[en]) * Ppap112v223[t]) * jxpulse[t] -
    GamATot112 * Ppap112v223[t]
ieqPAP112v223[neginf_, inival_] := Ppap112v223[neginf] == inival

```

Two-site PAPAs:

```

reqPAP221v123[en_, jxpulse_, truth_] := Ppap221v123'[t] == (fsig123v223[en] * Ppa221v223[t] -
    (fsigTot221[en] + fsigTot123[en, truth]) * Ppap221v123[t]) * jxpulse[t] -
    GamATot123 * Ppap221v123[t]
ieqPAP221v123[neginf_, inival_] := Ppap221v123[neginf] == inival

reqPAP203v123[en_, jxpulse_, truth_] := Ppap203v123'[t] == (fsig123v223[en] * Ppa203v223[t] -
    (fsigTot203[en] + fsigTot123[en, truth]) * Ppap203v123[t]) * jxpulse[t] -
    GamATot123 * Ppap203v123[t]
ieqPAP203v123[neginf_, inival_] := Ppap203v123[neginf] == inival

reqPAP212v123[en_, jxpulse_, truth_] := Ppap212v123'[t] == (fsig123v223[en] * Ppa212v223[t] -
    (fsigTot212[en] + fsigTot123[en, truth]) * Ppap212v123[t]) * jxpulse[t] -
    GamATot123 * Ppap212v123[t]
ieqPAP212v123[neginf_, inival_] := Ppap212v123[neginf] == inival

```

```
reqPAP[en_, jxpulse_] := Ppap'[t] == (fsig121v221[en] * Ppa221v223[t] + fsig103v203[en] *
  Ppa203v223[t] + fsig112v212[en] * Ppa212v223[t] + fsig123v223[en] * Ppa221v223[t] +
  fsig123v223[en] * Ppa203v223[t] + fsig123v223[en] * Ppa212v223[t]) * jxpulse[t]
ieqPAP[neginf_, inival_] := Ppap[neginf] == inival
```

```
reqPAPA[] := Ppapa'[t] == GamATot121 * Ppap121v223[t] + GamATot103 * Ppap103v223[t] +
  GamATot112 * Ppap112v223[t] + GamATot123 * (Ppap221v123[t] + Ppap203v123[t] +
  Ppap212v123[t])
ieqPAPA[neginf_, inival_] := Ppapa[neginf] == inival
```

Decay of SCHs:

```
reqSCHd[] := Pschd'[t] == GamATot123 * 2 * P223[t] * P123[t]
ieqSCHd[neginf_, inival_] := Pschd[neginf] == inival
```

Decay of ssDCHs:

```
reqssDCHd[] := Pssdchd'[t] == GamATot023 * 2 * P223[t] * P023[t]
ieqssDCHd[neginf_, inival_] := Pssdchd[neginf] == inival
```

Decay of tsDCHs:

```
reqtsDCHd[] := Ptsdchd'[t] == 2 * GamATot123 * P123[t]^2
ieqtsDCHd[neginf_, inival_] := Ptsdchd[neginf] == inival
```

Solve linear system of differential equations

```
papaSolveRateEqs[en_, jxfunc_, neginf_, posinf_, maxfrac_, truth_] :=
Module[{grndvec},
  grndvec = unitgrndvec[];
  NDSolve[{
    (* Molecular rate equations *)
    reqPA221v223[en, jxfunc], ieqPA221v223[neginf, 0],
    reqPA203v223[en, jxfunc], ieqPA203v223[neginf, 0],
    reqPA212v223[en, jxfunc], ieqPA212v223[neginf, 0],
    reqPAP121v223[en, jxfunc, truth], ieqPAP121v223[neginf, 0],
    reqPAP103v223[en, jxfunc, truth], ieqPAP103v223[neginf, 0],
    reqPAP112v223[en, jxfunc, truth], ieqPAP112v223[neginf, 0],
    reqPAP221v123[en, jxfunc, truth], ieqPAP221v123[neginf, 0],
    reqPAP203v123[en, jxfunc, truth], ieqPAP203v123[neginf, 0],
    reqPAP212v123[en, jxfunc, truth], ieqPAP212v123[neginf, 0],
    reqPAP[en, jxfunc], ieqPAP[neginf, 0],
    reqPAPA[], ieqPAPA[neginf, 0],
    reqSCHd[], ieqSCHd[neginf, 0],
    reqssDCHd[], ieqssDCHd[neginf, 0],
    reqtsDCHd[], ieqtsDCHd[neginf, 0],
```

```

(* Atomic rate equations *)
req223[en,jxfunc      ],ieq223[neginf,grndvec[[ 1]]],
req213[en,jxfunc      ],ieq213[neginf,grndvec[[ 2]]],
req222[en,jxfunc      ],ieq222[neginf,grndvec[[ 3]]],
req123[en,jxfunc,truth],ieq123[neginf,grndvec[[ 4]]],
req212[en,jxfunc      ],ieq212[neginf,grndvec[[ 5]]],
req203[en,jxfunc      ],ieq203[neginf,grndvec[[ 6]]],
req221[en,jxfunc      ],ieq221[neginf,grndvec[[ 7]]],
req113[en,jxfunc,truth],ieq113[neginf,grndvec[[ 8]]],
req122[en,jxfunc,truth],ieq122[neginf,grndvec[[ 9]]],
req023[en,jxfunc,truth],ieq023[neginf,grndvec[[10]]],
req202[en,jxfunc      ],ieq202[neginf,grndvec[[11]]],
req211[en,jxfunc      ],ieq211[neginf,grndvec[[12]]],
req220[en,jxfunc      ],ieq220[neginf,grndvec[[13]]],
req103[en,jxfunc,truth],ieq103[neginf,grndvec[[14]]],
req112[en,jxfunc,truth],ieq112[neginf,grndvec[[15]]],
req121[en,jxfunc,truth],ieq121[neginf,grndvec[[16]]],
req013[en,jxfunc,truth],ieq013[neginf,grndvec[[17]]],
req022[en,jxfunc,truth],ieq022[neginf,grndvec[[18]]],
req201[en,jxfunc      ],ieq201[neginf,grndvec[[19]]],
req210[en,jxfunc      ],ieq210[neginf,grndvec[[20]]],
req102[en,jxfunc,truth],ieq102[neginf,grndvec[[21]]],
req111[en,jxfunc,truth],ieq111[neginf,grndvec[[22]]],
req120[en,jxfunc,truth],ieq120[neginf,grndvec[[23]]],
req003[en,jxfunc,truth],ieq003[neginf,grndvec[[24]]],
req012[en,jxfunc,truth],ieq012[neginf,grndvec[[25]]],
req021[en,jxfunc,truth],ieq021[neginf,grndvec[[26]]],
req200[en,jxfunc      ],ieq200[neginf,grndvec[[27]]],
req101[en,jxfunc,truth],ieq101[neginf,grndvec[[28]]],
req110[en,jxfunc,truth],ieq110[neginf,grndvec[[29]]],
req011[en,jxfunc,truth],ieq011[neginf,grndvec[[30]]],
req020[en,jxfunc,truth],ieq020[neginf,grndvec[[31]]],
req002[en,jxfunc,truth],ieq002[neginf,grndvec[[32]]],
req100[en,jxfunc      ],ieq100[neginf,grndvec[[33]]],
req010[en,jxfunc,truth],ieq010[neginf,grndvec[[34]]],
req001[en,jxfunc,truth],ieq001[neginf,grndvec[[35]]],
req000[en,jxfunc      ],ieq000[neginf,grndvec[[36]]}],
{Ppa221v223[t],Ppa203v223[t],Ppa212v223[t],
Ppap121v223[t],Ppap103v223[t],Ppap112v223[t],
Ppap221v123[t],Ppap203v123[t],Ppap212v123[t],
Ppap[t],Ppapa[t],Pschd[t],Pssdchd[t],Ptsdchd[t],
P223[t],P213[t],P222[t],P123[t],P212[t],P203[t],P221[t],P113[t],
P122[t],P023[t],P202[t],P211[t],P220[t],P103[t],P112[t],P121[t],
P013[t],P022[t],P201[t],P210[t],P102[t],P111[t],P120[t],P003[t],
P012[t],P021[t],P200[t],P101[t],P110[t],P011[t],P020[t],P002[t],
P100[t],P010[t],P001[t],P000[t]},{t,neginf,posinf},
MaxStepFraction->maxfrac.MaxSteps->50000}

```

Private section

```
Begin["`Private`"]
```

Unit conversions

```
BohrToAngstroem:=0.5291772083  
AngstroemToBohr:=1/BohrToAngstroem  
MicrometerToBohr:=1*^4*AngstroemToBohr  
FemtoSecondTot0:=1/24.18884326505*^-3
```

Terminate package

```
End[]
```

```
EndPackage[]
```

References

- [1] Nina Rohringer, Robin Santra, *Phys. Rev. A* 76, 033416 (2007), <http://dx.doi.org/10.1103/PhysRevA.76.033416>
- [2] Los Alamos National Laboratory Atomic Physics Codes, <http://aphysics2.lanl.gov/cgi-bin/ION/runlanl08a.pl>
- [3] LCLS Parameter Index for X-RAY-OPTICS / RADIATION-SOURCE, http://www-ssrl.slac.stanford.edu/htbin/rdbweb/LCLS_params_DB_public
- [4] M. G. Makris, P. Lambropoulos, A. Mihelič, *Phys. Rev. Lett.* 102, 033002 (2009), <http://dx.doi.org/10.1103/PhysRevLett.102.033002>
- [5] J. L. Campbell, Tibor Papp, Widths of the atomic K–N7 levels, *At. Data Nucl. Data Tables* 77, 1–56 (2001), <http://dx.doi.org/10.1006/adnd.2000.0848>
- [6] Ryan Coffee, John Bozek, Christian Buth, Marcus Hertlein, Markus Gühr, Nora Berrah, Ali Belkacem, Linda Young, Louis DiMauro, Hamed Merdji, Janos Hajdu, Philip H. Bucksbaum, Xray multiple ionization of impulsively aligned molecules, Linac Coherent Light Source, Menlo Park, California, USA, 2009, research proposal for the first round of beam time at the Linac Coherent Light Source
- [7] Mau Hsiung Chen, *Phys. Rev. A* 44, 239-242 (1991), <http://dx.doi.org/10.1103/PhysRevA.44.239>

- [8] Albert C. Thompson et al., X-ray data booklet, Center for X-ray Optics and Advanced Light Source, Lawrence Berkeley National Laboratory, <http://xdb.lbl.gov>
- [9] Christian Buth, Robin Santra, Phys. Rev. A 78, 043409 (2008), <http://dx.doi.org/10.1103/PhysRevA.78.043409>
- [10] W. C. Stolte, Z. X. He, J. N. Cutler, Y. Lu, J. A. R. Samson, At. Data Nucl. Data Tables 69, 171-179 (1998), <http://dx.doi.org/10.1006/adnd.1998.0775>